

C2M3

Simpson's and Trapezoidal Rule

Riemann sums, Simpson's Rule, and the Trapezoidal Rule are available in Maple in the Student package. The example chosen here involves the sine function on the interval $[1, 3]$ using 60 subintervals for the sums and 10 subintervals for the graphics. Because we wanted the decimal or floating point answer we used `evalf` instead of `value`, which would have listed a long summation. The actual integral is $\int_1^3 \sin x dx = -\cos 3 + \cos 1 \approx 1.530294803$. Please observe the output of each command line below.

```
> restart: with(student):
> simpson(sin(x),x=1..3,60);

$$\frac{1}{90} \sin(1) + \frac{1}{90} \sin(3) + \frac{2}{45} \left( \sum_{i=1}^{30} \sin \left( \frac{29}{30} + \frac{1}{15} i \right) \right) + \frac{1}{45} \left( \sum_{i=1}^{29} \sin \left( 1 + \frac{1}{15} i \right) \right)$$

> evalf(%);
1.530294813
> trapezoid(sin(x),x=1..3,60);

$$\frac{1}{60} \sin(1) + \frac{1}{30} \left( \sum_{i=1}^{59} \sin \left( 1 + \frac{1}{30} i \right) \right) + \frac{1}{60} \sin(3)$$

> evalf(%);
1.530153105
> Int(sin(x),x=1..3);

$$\int_1^3 \sin(x) dx$$

> value(%);

$$-\cos(3) + \cos(1)$$

> evalf(%);
1.530294803
```

Accuracy The error estimates for the Trapezoidal Rule and Simpson's Rule are stated in the course textbook. As a reminder, if $|f''(x)| \leq K$ and $|f^{(iv)}(x)| \leq M$, and n subintervals are used, then the errors for the respective rules, E_T and E_S , satisfy

$$|E_T| \leq \frac{K(b-a)^3}{12n^2} \quad \text{and} \quad |E_S| \leq \frac{M(b-a)^5}{180n^4}$$

when applied over the interval $[a, b]$.

Trapezoidal Rule Maple Example Approximate $\int_0^{\pi/3} \sin(2x) dx$ to within $\frac{1}{1000}$ using the Trapezoidal Rule. Determine the number of subintervals necessary to achieve the requested accuracy by applying the estimate displayed above. It is best if our second derivative is a function, so we use `unapply` to create a function from an expression.

```
> restart: with(student):
> f:=x->sin(2*x);
f := x → sin(2x)
> f2:=unapply(diff(f(x),x,x),x);
f2 := x → -4 sin(2x)
```

At this point we have the second derivative of f as a function. We must find the maximum value of the absolute value of the second derivative on the interval.

```
> K:=maximize(abs(f2(x)),x,{x=0..evalf(Pi/3)});
```

$$K := 4$$

Equate the error and the overestimate and solve for the value of n that works.

```
> Eqn1:=(Pi/3-0)^3*K/(12*n^2)=1/1000;
```

$$Eqn1 := \frac{1}{81} \frac{\pi^3}{n^2} = \frac{1}{1000}$$

```
> solve(Eqn1,n);
```

$$\frac{10}{9} \sqrt{10} \pi^{(3/2)}, -\frac{10}{9} \sqrt{10} \pi^{(3/2)}$$

```

> evalf(%);
19.56511025, -19.56511025
Since  $n$  must be an integer, choose  $n = 20$ 
> app:=trapezoid(f(x),x=0..Pi/3,20);

$$app := \frac{1}{120}\pi \left( 2 \left( \sum_{i=1}^{19} \sin\left(\frac{1}{30}i\pi\right) \right) + \frac{1}{2}\sqrt{3} \right)$$

> approx:=evalf(app);
approx := .7493144853
> ans:=evalf(int(f(x),x=0..Pi/3));
ans := .7500000000
> abs(approx-ans);
.0006855147

```

So, we have achieved the requested accuracy.

C2M3 Problems:

1. Use Maple to find the requested approximations. $\int_0^2 \sqrt{1+x^2} dx$, $n = 40$, use `simpson`, `trapezoid`
2. Modify the Maple Example above and use Simpson's Rule instead of the Trapezoidal Rule to approximate $\int_0^2 \frac{x^2}{1+x^4} dx$ to within $\frac{1}{1000}$.
3. Modify the Maple Example above and then use Simpson's Rule instead of the Trapezoidal Rule to approximate $\int_0^1 x \arctan(x) dx$ to within $\frac{1}{10000}$.